
20 Expression Optimization

Previous chapters discussed optimization over a fixed set of design variables. For
many problems, the number of variables is unknown, such as in the optimization
of graphical structures or computer programs. Designs in these contexts can be
represented by expressions that belong to a grammar. This chapter discusses
ways to make the search of optimal designs more efficient by accounting for the
grammatical structure of the design space.

20.1 Grammars

An expression can be represented by a tree of symbols. For example, themathemati-
cal expression x+ ln 2 can be represented using the tree in figure 20.1 consisting of
the symbols +, x, ln, and 2. Grammars specify constraints on the space of possible
expressions.

+

x ln

2

Figure 20.1. The expression x +
ln 2 represented as a tree.

A grammar is represented by a set of production rules. These rules involve
symbols as well as types. A type can be interpreted as a set of expression trees. A
production rule represents a possible expansion of type to an expression involving
symbols or types. If a rule expands only to symbols, then it is called terminal
because it cannot be expanded further. An example of a nonterminal rule is
R 7→ R + R, which means that the type R can consist of elements of the set R

added to elements in the set R.1 1 This chapter focuses on context-
free grammars, but other forms exist.
See L. Kallmeyer, Parsing Beyond
Context-Free Grammars. Springer,
2010.

We can generate an expression from a grammar by starting with a start type
and then recursively applying different production rules. We stop when the
tree contains only symbols. Figure 20.2 illustrates this process for the expression
x + ln 2. An application to natural language expressions is shown in example 20.1.

The number of possible expressions allowed by a grammar can be infinite.
Example 20.2 shows a grammar that allows for infinitely many valid expressions.

Figure 20.2. Using the production
rules

R 7→ R + R

R 7→ x

R 7→ ln(R)

R 7→ 2

to generate x + ln 2. Blue nodes are
unexpanded types.

Consider a grammar that allows for the generation of simple English state-
ments:

S 7→ N V

V 7→ V A

A 7→ rapidly | efficiently
N 7→ Alice | Bob | Mykel | Tim
V 7→ runs | reads | writes

The types S, N, V, and A correspond to statements, nouns, verbs, and
adverbs, respectively. An expression is generated by starting with the type S

and iteratively replacing types:

S

N V

Mykel V A

Mykel writes rapidly

Not all terminal symbol categories must be used. For instance, the statement
‘‘Alice runs’’ can also be generated.

Example 20.1. A grammar for pro-
ducing simple English statements.
Using | on the right-hand side of
an expression is shorthand for or.
Thus, the rule

A 7→ rapidly | efficiently
is equivalent to having two rules,
A 7→ rapidly and A 7→ efficiently.

20.1 . grammars 363

Expression optimization often constrains expressions to a maximum depth or
penalizes expressions based on their depth or node count. Even if the grammar
allows a finite number of expressions, the space is often too vast to search exhaus-
tively. Hence, there is a need for algorithms that efficiently search the space of
possible expressions for one that optimizes an objective function.

Consider a four-function calculator grammar that applies addition, subtrac-
tion, multiplication, and division to the ten digits:

R 7→ R + R

R 7→ R−R

R 7→ R×R

R 7→ R / R

R 7→ 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9

An infinite number of expressions can be generated because the nonterminal
R can always be expanded into one of the calculator operations.

Many expressions will produce the same value. Addition and multipli-
cation operators are commutative, meaning that the order does not matter.
For example, a + b is the same as b + a. These operations are also associative,
meaning the order in which multiple operations of the same type occur do
not matter. For example, a× b× c is the same as c× b× a. Other operations
preserve values, like adding zero or multiplying by one.

Not all expressions under this grammar are mathematically valid. For
example, division by zero is undefined. Removing zero as a terminal symbol
is insufficient to prevent this error because zero can be constructed using
other operations, such as 1− 1. Such exceptions are often handled by the
objective function, which can catch exceptions and penalize them.

Example 20.2. Some of the chal-
lenges associated with grammars,
as illustrated with a four-function
calculator grammar.

The expression optimization routines covered in the chapter use ExprRules.jl.
Grammars can be defined using the grammar macro by listing the production
rules, as shown in example 20.3.

Many of the expression optimization algorithms involve manipulating com-
ponents of an expression tree in a way that preserves the way the types were
expanded. A RuleNode object tracks which production rules were applied when

We may define a grammar using the grammar macro. The nonterminals are
on the left of the equal sign, and the expressions with terminals and nonter-
minals are the on the right. The package includes some syntax to represent
grammars more compactly.
using ExprRules
grammar = @grammar begin

R = x # reference a variable
R = R * A # multiple children
R = f(R) # call a function
R = _(randn()) # random variable generated on node creation
R = 1 | 2 | 3 # equivalent to R = 1, R = 2, and R = 3
R = |(4:6) # equivalent to R = 4, R = 5, and R = 6
A = 7 # rules for different return types

end;

Example 20.3. Example of
defining a grammar using the
ExprRules.jl package.

doing an expansion. Calling rand with a specified starting type will generate
a random expression represented by a RuleNode. Calling sample will select a
random RuleNode from an existing RuleNode tree. Nodes are evaluated using
eval.

Themethod return_type returns the node’s return type as a symbol, isterminal
returns whether the symbol is terminal, child_types returns the list of nontermi-
nal symbols associated with the node’s production rule, and nchildren returns
the number of children. These four methods each take as input the grammar
and the node. The number of nodes in an expression tree is obtained using
length(node), and the depth is obtained using depth(node).

A third type, NodeLoc, is used to refer to a node’s location in the expression
tree. Subtrees manipulation often requires NodeLocs.
loc = sample(NodeLoc, node); # uniformly sample a node loc
loc = sample(NodeLoc, node, :R, grammar); # sample a node loc of type R
subtree = get(node, loc);

20.2 Genetic Programming

Genetic algorithms (see chapter 9) use chromosomes that encode design points
in a sequential format. Genetic programming2 represents individuals using trees

2 J. R. Koza, Genetic Programming:
On the Programming of Computers
by Means of Natural Selection. MIT
Press, 1992.

20.2. genetic programming 365

instead (figure 20.3), which are better at representing mathematical functions,
programs, decision trees, and other hierarchical structures.

if

≤

x 2

×

x +

y 3

0

Figure 20.3. A tree representation
of the Julia method:
x ≤ 2 ? x*(y+3) : 0

Similar to genetic algorithms, genetic programs are initialized randomly and
support crossover and mutation. In tree crossover (figure 20.4), two parent trees
are mixed to form a child tree. A random node is chosen in each parent, and the
subtree at the chosen node in the first parent is replaced with the subtree at the
chosen node of the second parent. Tree crossover works on parents with different
sizes and shapes, allowing arbitrary trees to mix. In some cases one must ensure
that replacement nodes have certain types, such as Boolean values input into the
condition of an if statement.3 Tree crossover is implemented in algorithm 20.1.

3 This book focuses only on ge-
netic operations that adhere to
the constraints of the grammar.
Sometimes genetic programming
with this restriction is referred to
as strongly typed genetic program-
ming, as discussed in D. J. Mon-
tana, “Strongly Typed Genetic Pro-
gramming,” Evolutionary Computa-
tion, vol. 3, no. 2, pp. 199–230, 1995.

Parent A Parent B Child

Figure 20.4. Tree crossover is used
to combine two parent trees to pro-
duce a child tree.

Tree crossover tends to produce trees with greater depth than the parent trees.
Each generation tends to increase in complexity, which often results in overly
complicated solutions and slower runtimes.We encourage parsimony, or simplicity,
in the solution, by introducing a small bias in the objective function value based
on a tree’s depth or node count.

Applying tree mutation (figure 20.5) starts by choosing a random node in the
tree. The subtree rooted at that node is deleted, and a new random subtree is
generated to replace the old subtree. In contrast to mutation in binary chromo-

struct TreeCrossover <: CrossoverMethod
grammar
max_depth

end
function crossover(C::TreeCrossover, a, b)

child = deepcopy(a)
crosspoint = sample(b)
typ = return_type(C.grammar, crosspoint.ind)
d_subtree = depth(crosspoint)
d_max = C.max_depth + 1 - d_subtree
if d_max > 0 && contains_returntype(child,C.grammar,typ,d_max)

loc = sample(NodeLoc, child, typ, C.grammar, d_max)
insert!(child, loc, deepcopy(crosspoint))

end
child

end

Algorithm 20.1. Tree crossover
implemented for a and b of type
RuleNode from ExprRules.jl. The
TreeCrossover struct contains a
rule set grammar and a maximum
depth max_depth.

somes, treemutation can typically occur at most once, oftenwith a low probability
around 1%. Tree mutation is implemented in algorithm 20.2.

Before After

Figure 20.5. Tree mutation deletes
a random subtree and generates a
new one to replace it.

struct TreeMutation <: MutationMethod
grammar
p

end
function mutate(M::TreeMutation, a)

child = deepcopy(a)
if rand() < M.p

loc = sample(NodeLoc, child)
typ = return_type(M.grammar, get(child, loc).ind)
subtree = rand(RuleNode, M.grammar, typ)
insert!(child, loc, subtree)

end
return child

end

Algorithm 20.2. Tree mutation im-
plemented for an individual a of
type RuleNode from ExprRules.jl.
The TreeMutation struct contains
a rule set grammar and a mutation
probability p.

Tree permutation (figure 20.6) is a second form of genetic mutation. Here, the
children of a randomly chosen node are randomly permuted. Tree permutation
alone is typically not sufficient to introduce new genetic material into a population
and is often combined with tree mutation. Tree permutation is implemented in
algorithm 20.3.

Before After

Figure 20.6. Tree permutation per-
mutes the children of a randomly
chosen node.

The implementation of genetic programming is otherwise identical to that
of genetic algorithms. More care must typically be taken in implementing the
crossover and mutation routines, particularly when determining what sorts of
nodes can be generated and that only syntactically correct trees are produced.
Genetic programming is used to generate an expression that approximates π in
example 20.4.

struct TreePermutation <: MutationMethod
grammar
p

end
function mutate(M::TreePermutation, a)

child = deepcopy(a)
if rand() < M.p

node = sample(child)
n = length(node.children)
types = child_types(M.grammar, node)
for i in 1 : n-1

c = 1
for k in i+1 : n

if types[k] == types[i] &&
rand() < 1/(c+=1)

node.children[i], node.children[k] =
node.children[k], node.children[i]

end
end

end
end
return child

end

Algorithm 20.3. Tree permutation
implemented for an individual a of
type RuleNode from ExprRules.jl,
where p is the mutation probabil-
ity.

Consider approximating π using only operations on a four-function calcula-
tor. We can solve this problem using genetic programming where nodes can
be any of the elementary operations: add, subtract, multiply, and divide, and
the digits 1− 9.

We use ExprRules.jl to specify our grammar:
grammar = @grammar begin

R = |(1:9)
R = R + R
R = R - R
R = R / R
R = R * R

end

We construct an objective function and penalize large trees:
function f(node)

value = eval(node, grammar)
if isinf(value) || isnan(value)

return Inf
end
Δ = abs(value - π)
return log(Δ) + length(node)/1e3

end

We finally run our genetic program using a call to the genetic_algorithm
function from section 9.2:
population = [rand(RuleNode, grammar, :R) for i in 1:1000]
best_tree = genetic_algorithm(f, population, 30,

TruncationSelection(50),
TreeCrossover(grammar, 10),
TreeMutation(grammar, 0.25))

The best performing tree is shown on the right. It evaluates to 3.141586, which
matches π to four decimal places.

Example 20.4. Using genetic pro-
gramming to estimate π using only
digits and the four principle arith-
metic operations.

20.3 Grammatical Evolution

Grammatical evolution4 operates on an integer array instead of a tree, allowing the 4 C. Ryan, J. J. Collins, and M.O.
Neill, “Grammatical Evolution:
Evolving Programs for an Ar-
bitrary Language,” in European
Conference on Genetic Programming,
1998.

same techniques employed in genetic algorithms to be applied. Unlike genetic
algorithms, the chromosomes in grammatical evolution encode expressions based
on a grammar. Grammatical evolution was inspired by genetic material, which is
inherently serial like the chromosomes used in genetic algorithms. 5 5 Our serial DNA is read and used

to construct complicated protein
structures. DNA is often referred
to as the genotype—the object
on which genetic operations are
performed. The protein structure
is the phenotype—the object en-
coded by the genotype whose per-
formance is evaluated. The gram-
matical evolution literature often
refers to the integer design vector
as the genotype and the resulting
expression as the phenotype.

In grammatical evolution, designs are integer arrays much like the chromo-
somes used in genetic algorithms. Each integer is unbounded because indexing
is performed using modular arithmetic. The integer array can be translated into
an expression tree by parsing it from left to right.

We begin with a starting symbol and a grammar. Suppose n rules in the
grammar can be applied to the starting symbol. The jth rule is applied, where
j = i mod1 n and i is the first integer in the integer array.6

6 We use x mod1 n to refer to the 1-
index modulus:

((x− 1) mod n) + 1

This type ofmodulus is useful with
1-based indexing. The correspond-
ing Julia function is mod1.

We then consider the rules applicable to the resulting expression and use
similar modular arithmetic based on the second integer in the array to select
which rule to apply. This process is repeated until no rules can be applied and the
phenotype is complete.7 The decoding process is implemented in algorithm 20.4

7 No genetic information is read
when there is only a single applica-
ble rule.

and is worked through in example 20.5.
It is possible for the integer array to be too short, thereby causing the translation

process to run past the length of the array. Rather than producing an invalid
individual and penalizing it in the objective function, the process wraps around
to the beginning of the array instead. This wrap-around effect means that the same
decision can be read several times during the transcription process. Transcription
can result in infinite loops, which can be prevented by a maximum depth.

Genetic operations work directly on the integer design array. We can adopt
the operations used on real-valued chromosomes and apply them to the integer-
valued chromosomes. The only change is that mutation must preserve real values.
A mutation method for integer-valued chromosomes using zero-mean Gaussian
perturbations is implemented in algorithm 20.5.

Grammatical evolution uses two additional genetic operators. The first, gene
duplication, occurs naturally as an error in DNA replication and repair. Gene
duplication can allow new genetic material to be generated and can store a second
copy of a useful gene to reduce the chance of a lethal mutation removing the gene
from the gene pool. Gene duplication chooses a random interval of genes in the

struct DecodedExpression
node
n_rules_applied

end
function decode(x, grammar, sym, c_max=1000, c=0)

node, c = _decode(x, grammar, sym, c_max, c)
DecodedExpression(node, c)

end
function _decode(x, grammar, typ, c_max, c)

types = grammar[typ]
if length(types) > 1

g = x[mod1(c+=1, length(x))]
rule = types[mod1(g, length(types))]

else
rule = types[1]

end
node = RuleNode(rule)
childtypes = child_types(grammar, node)
if !isempty(childtypes) && c < c_max

for ctyp in childtypes
cnode, c = _decode(x, grammar, ctyp, c_max, c)
push!(node.children, cnode)

end
end
return (node, c)

end

Algorithm 20.4. A method for de-
coding an integer design vector
to produce an expression, where
x is a vector of integers, grammar
is a Grammar, and sym is the root
symbol. The counter c is used
during the recursion process and
the parameter c_max is an upper
limit on the maximum number of
rule applications, to prevent an in-
finite loop. The method returns
a DecodedExpression, which con-
tains the expression tree and the
number of rules applied during the
decoding process.

Consider a grammar for real-valued strings:

R 7→ D D′ P E

D′ 7→ D D′ | ǫ
P 7→ . D D′ | ǫ
E 7→ E S D D′ | ǫ
S 7→ + | − | ǫ

D 7→ 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9

where R is a real value, D is a terminal decimal, D′ is a nonterminal decimal,
P is the decimal part, E is the exponent, and S is the sign. Any ǫ values
produce empty strings.

Suppose our design is [205, 52, 4, 27, 10, 59, 6] and we have the starting
symbol R. There is only one applicable rule, so we do not use any genetic
information and we replace R with DD′PE.

Next we must replace D. There are 10 options. We select 205 mod1 10 = 5,
and thus obtain 4D′PE

Next we replace D′, which has two options.We select index 52 mod1 2 = 2,
which corresponds to ǫ.

Continuing in this manner we produce the string 4E+8.
The above grammar can be implemented in ExprRules using:
grammar = @grammar begin

R = D * De * P * E
De = D * De | ""
P = "." * D * De | ""
E = "E" * S * D * De | ""
S = "+"|"-"|""
D = "0"|"1"|"2"|"3"|"4"|"5"|"6"|"7"|"8"|"9"

end

and can be evaluated using:
x = [205, 52, 4, 27, 10, 59, 6]
str = eval(decode(x, grammar, :R).node, grammar)

Example 20.5. The process by
which an integer design vector in
grammatical evolution is decoded
into an expression.

Our implementation is depth-
first. If 52 were instead 51, the rule
D′ 7→ D D′ would be applied,
followed by selecting a rule for
the new D, eventually resulting in
43950.950E+8.

struct IntegerGaussianMutation <: MutationMethod
σ

end
function mutate(M::IntegerGaussianMutation, child)

return child + round.(Int, randn(length(child)).*M.σ)
end

Algorithm 20.5. The Gaussian mu-
tation method modified to pre-
serve integer values for integer-
valued chromosomes. Each value
is perturbed by a zero-mean Gaus-
sian random value with standard
deviation σ and then rounded to
the nearest integer.

chromosome to duplicate. A copy of the selected interval is appended to the back
of the chromosome. Duplication is implemented in algorithm 20.6.

struct GeneDuplication <: MutationMethod
end
function mutate(M::GeneDuplication, child)

n = length(child)
i, j = rand(1:n), rand(1:n)
interval = min(i,j) : max(i,j)
return vcat(child, deepcopy(child[interval]))

end

Algorithm 20.6. The gene duplica-
tion method used in grammatical
evolution.

The second genetic operation, pruning, tackles a problem encountered during
crossover. As illustrated in figure 20.7, crossover will select a crossover point at
random in each chromosome and construct a new chromosome using the left
side of the first and the right side of the second chromosome. Unlike genetic
algorithms, the trailing entries in chromosomes of grammatical evolution may
not be used; during parsing, once the tree is complete, the remaining entries
are ignored. The more unused entries, the more likely it is that the crossover
point lies in the inactive region, thus not providing new beneficial material. An
individual is pruned with a specified probability, and, if pruned, its chromosome
is truncated such that only active genes are retained. Pruning is implemented in
algorithm 20.7 and is visualized in figure 20.8.

Like genetic programming, grammatical evolution can use the genetic algo-
rithm method.8 We can construct a MutationMethod that applies multiple mu- 8 Genotype to phenotype mapping

would occur in the objective func-
tion.tation methods in order to use pruning, duplication, and standard mutation

approaches in tandem. Such a method is implemented in algorithm 20.8.
Grammatical evolution suffers from two primary drawbacks. First, it is difficult

to tell whether the chromosome is feasible without decoding it into an expression.

parent a

parent b

child

Figure 20.7. Crossover applied to
chromosomes in grammatical evo-
lution may not affect the active
genes in the front of the chromo-
some. The child shown here inher-
its all of the active shaded genes
from parent a so it will effectively
act as an identical expression. Prun-
ing was developed to overcome
this issue.

struct GenePruning <: MutationMethod
p
grammar
typ

end
function mutate(M::GenePruning, child)

if rand() < M.p
c = decode(child, M.grammar, M.typ).n_rules_applied
if c < length(child)

child = child[1:c]
end

end
return child

end

Algorithm 20.7. The gene pruning
method used in grammatical evo-
lution.

before

after

Figure 20.8. Pruning truncates the
chromosome such that only active
genes remain.

struct MultiMutate <: MutationMethod
Ms

end
function mutate(M::MultiMutate, child)

for m in M.Ms
child = mutate(m, child)

end
return child

end

Algorithm 20.8. A
MutationMethod for applying
all mutation methods stored in the
vector Ms.

Second, a small change in the chromosome may produce a large change in the
corresponding expression.

20.4 Probabilistic Grammars

A probabilistic grammar9 adds a weight to each rule in the genetic program’s gram- 9 T. L. Booth and R.A. Thompson,
“Applying Probability Measures to
Abstract Languages,” IEEE Transac-
tions on Computers, vol. C-22, no. 5,
pp. 442–450, 1973.

mar. When sampling from all applicable rules for a given node, we select a rule
stochastically according to the relative weights. The probability of an expression is
the product of the probabilities of sampling each rule. Algorithm 20.9 implements
the probability calculation. Example 20.6 demonstrates sampling an expression
from a probabilistic grammar and computes its likelihood.

struct ProbabilisticGrammar
grammar
ws

end
function probability(probgram, node)

typ = return_type(probgram.grammar, node)
i = findfirst(isequal(node.ind), probgram.grammar[typ])
prob = probgram.ws[typ][i] / sum(probgram.ws[typ])
for (i,c) in enumerate(node.children)

prob *= probability(probgram, c)
end
return prob

end

Algorithm 20.9. Amethod for com-
puting the probability of an expres-
sion based on a probabilistic gram-
mar, where probgram is a proba-
bilistic grammar consisting of a
grammar grammar and a mapping
of types to weights for all applica-
ble rules ws, and node is a RuleNode
expression.

Optimization using a probabilistic grammar improves its weights with each
iteration using elite samples from a population. At each iteration, a population of
expressions is sampled and their objective function values are computed. Some
number of the best expressions are considered the elite samples and can be used
to update the weights. A new set of weights is generated for the probabilistic
grammar, where the weight wT

i for the ith production rule applicable to return
type T is set to the number of times the production rule was used in generating
the elite samples. This update procedure is implemented in algorithm 20.10.

The probabilistic grammars above can be extended to more complicated proba-
bility distributions that consider other factors, such as the depth in the expression
or local dependencies among siblings in subtrees. One approach is to use Bayesian
networks.10

10 P.K. Wong, L. Y. Lo, M. L. Wong,
and K. S. Leung, “Grammar-
Based Genetic Programming
with Bayesian Network,” in
IEEE Congress on Evolutionary
Computation (CEC), 2014.

Consider a probabilistic grammar for strings composed entirely of ‘‘a’’s:

A 7→ a A wA
1 = 1

7→ a B a A wA
2 = 3

7→ ǫ wA
3 = 2

B 7→ a B wB
1 = 4

7→ ǫ wB
2 = 1

where we have a set of weights w for each parent type and ǫ is an empty
string.

Suppose we generate an expression starting with the type A. The proba-
bility distribution over the three possible rules is:

P(A 7→ a A) = 1/(1 + 3 + 2) = 1/6

P(A 7→ a B a A) = 3/(1 + 3 + 2) = 1/2

P(A 7→ ǫ) = 2/(1 + 3 + 2) = 1/3

Suppose we sample the second rule and obtain a B a A.
Next we sample a rule to apply to B. The probability distribution over the

two possible rules is:

P(B 7→ a B) = 4/(4 + 1) = 4/5

P(B 7→ ǫ) = 1/(4 + 1) = 1/5

Suppose we sample the second rule and obtain a ǫ a A.
Next we sample a rule to apply to A. Suppose we sample A 7→ ǫ to obtain

a ǫ a ǫ, which produces the ‘‘a’’-string ‘‘aa’’. The probability of the sequence
of rules applied to produce ‘‘aa’’ under the probabilistic grammar is:

P(A 7→ a B a A)P(B 7→ ǫ)P(A 7→ ǫ) =
1

2
· 1

5
· 1

3
=

1

30

Note that this is not the same as the probability of obtaining ‘‘aa’’, as other
sequences of production rules could also have produced it.

Example 20.6. Sampling an expres-
sion from a probabilistic grammar
and computing the expression’s
likelihood.

function _update!(probgram, x)
grammar = probgram.grammar
typ = return_type(grammar, x)
i = findfirst(isequal(x.ind), grammar[typ])
probgram.ws[typ][i] += 1
for c in x.children

_update!(probgram, c)
end
return probgram

end
function update!(probgram, Xs)

for w in values(probgram.ws)
fill!(w,0)

end
for x in Xs

_update!(probgram, x)
end
return probgram

end

Algorithm 20.10. A method for ap-
plying a learning update to a prob-
abilistic grammar probgram based
on an elite sample of expressions
Xs.

20.5 Probabilistic Prototype Trees

The probabilistic prototype tree11 is a different approach that learns a distribution 11 R. Salustowicz and J. Schmidhu-
ber, “Probabilistic Incremental Pro-
gramEvolution,”Evolutionary Com-
putation, vol. 5, no. 2, pp. 123–141,
1997.

for every node in the expression tree. Each node in a probabilistic prototype
tree contains a probability vector representing a categorical distribution over
the grammar’s production rules. The probability vectors are updated to reflect
knowledge gained from successive generations of expressions. The maximum
number of children for a node is the maximum number of nonterminals among
rules in the grammar.12 12 The arity of a function is the

number of arguments. The arity
of a grammar rule, which can be
viewed as a function, is the num-
ber of nonterminals in the rule.

Probability vectors are randomly initialized when a node is created. Random
probability vectors can be drawn from a Dirichlet distribution.13 The original im-

13 The Dirichlet distribution is of-
ten used to represent a distribu-
tion over discrete distributions. D.
Barber, Bayesian Reasoning and Ma-
chine Learning. Cambridge Univer-
sity Press, 2012.

plementation initializes terminals to a scalar value of 0.6 and nonterminals to 0.4.
In order to handle strongly-typed grammars we maintain a probability vector for
applicable rules to each parent type. Algorithm 20.11 defines a node type and
implements this initialization method.

Expressions are sampled using the probability vectors in the probabilistic
prototype tree. A rule in a node is drawn from the categorical distribution defined
by the node’s probability vector for the required return type, normalizing the

struct PPTNode
ps
children

end
function PPTNode(grammar;

w_terminal = 0.6,
w_nonterm = 1-w_terminal,
)

ps = Dict(typ => normalize!([isterminal(grammar, i) ?
w_terminal : w_nonterm
for i in grammar[typ]], 1)

for typ in nonterminals(grammar))
PPTNode(ps, PPTNode[])

end
function get_child(ppt::PPTNode, grammar, i)

if i > length(ppt.children)
push!(ppt.children, PPTNode(grammar))

end
return ppt.children[i]

end

Algorithm 20.11. A probabilistic
prototype tree node type and asso-
ciated initialization functionwhere
ps is a dictionary mapping a sym-
bol corresponding to a return type
to a probability vector over applica-
ble rules, and children is a list of
PPTNodes. The method get_child
will automatically expand the tree
when attempting to access a non-
existent child.

associated probability vector values to obtain a valid probability distribution. The
tree is traversed in depth-first order. This sampling procedure is implemented in
algorithm 20.12 and visualized in figure 20.9.

Learning can use information either from an entire sampled population or
from elite samples. Let the best expression in the current generation be xbest and
the best expression found so far be xelite. The node probabilities are updated to
increase the likelihood of generating xbest.14 14 The original probabilistic proto-

type tree implementation will peri-
odically increase the likelihood of
generating xelite.

The probability of generating xbest is the product of the probabilities of choosing
each rule in xbest when traversing through the probabilistic prototype tree. We
compute a target probability for P(xbest):

Ptarget = P(xbest) + (1− P(xbest)) · α ·
ǫ− yelite
ǫ− ybest

(20.1)

where α and ǫ are positive constants. The fraction on the right-hand side produces
larger steps toward expressions with better objective function values. The target
probability can be calculated using algorithm 20.13.

function rand(ppt, grammar, typ)
rules = grammar[typ]
rule_index = sample(rules, Weights(ppt.ps[typ]))
ctypes = child_types(grammar, rule_index)

arr = Vector{RuleNode}(undef, length(ctypes))
node = iseval(grammar, rule_index) ?

RuleNode(rule_index, eval(grammar, rule_index), arr) :
RuleNode(rule_index, arr)

for (i,typ) in enumerate(ctypes)
node.children[i] =

rand(get_child(ppt, grammar, i), grammar, typ)
end
return node

end

Algorithm 20.12. A method for
sampling an expression from a
probabilistic prototype tree. The
tree is expanded as needed.

p1

p1

p11

p111

p12

p1

p11 p12

p111 p112 p121 p122

×
ln

x

2

sample 1st expression
×

− +

3 x 2 7

sample 2nd expression

Figure 20.9. A probabilistic pro-
totype tree initially contains only
the root node but expands as ad-
ditional nodes are needed during
expression generation.

function probability(ppt, grammar, expr)
typ = return_type(grammar, expr)
i = findfirst(isequal(expr.ind), grammar[typ])
p = ppt.ps[typ][i]
for (i,c) in enumerate(expr.children)

p *= probability(get_child(ppt, grammar, i), grammar, c)
end
return p

end
function p_target(ppt, grammar, x_best, y_best, y_elite, α, ϵ)

p_best = probability(ppt, grammar, x_best)
return p_best + (1-p_best)*α*(ϵ - y_elite)/(ϵ - y_best)

end

Algorithm 20.13. Methods for com-
puting the probability of an ex-
pression and the target probabil-
ity, where ppt is the root node
of the probabilistic prototype tree,
grammar is the grammar, expr and
x_best are RuleNode expressions,
y_best and y_elite are scalar ob-
jective function values, and α and
ϵ are scalar parameters.

The target probability is used to adjust the probability vectors in the proba-
bilistic prototype tree. The probabilities associated with the chosen nodes are
increased iteratively until the target probability is exceeded:

P
(

x
(i)
best
)

← P
(

x
(i)
best
)

+ c · α ·
(

1− P
(

x
(i)
best
))

for all i in {1, 2, . . .} (20.2)

where x
(i)
best is the ith rule applied in expression xbest and c is a scalar.15 15 A recommended value is c = 0.1.

The adapted probability vectors are then renormalized to 1 by downscaling the
values of all nonincreased vector components proportionally to their current value.
The probability vector p, where the ith component was increased, is adjusted
according to:

pj ← pj
1− pi

‖p‖1 − pi
for j 6= i (20.3)

The learning update is implemented in algorithm 20.14.

function _update!(ppt, grammar, x, c, α)
typ = return_type(grammar, x)
i = findfirst(isequal(x.ind), grammar[typ])
p = ppt.ps[typ]
p[i] += c*α*(1-p[i])
psum = sum(p)
for j in 1 : length(p)

if j != i
p[j] *= (1- p[i])/(psum - p[i])

end
end
for (pptchild,xchild) in zip(ppt.children, x.children)

_update!(pptchild, grammar, xchild, c, α)
end
return ppt

end
function update!(ppt, grammar, x_best, y_best, y_elite, α, c, ϵ)

p_targ = p_target(ppt, grammar, x_best, y_best, y_elite, α, ϵ)
while probability(ppt, grammar, x_best) < p_targ

_update!(ppt, grammar, x_best, c, α)
end
return ppt

end

Algorithm 20.14. A method for
applying a learning update to a
probabilistic prototype tree with
root ppt, grammar grammar, best
expression x_best with objective
function value y_best, elite objec-
tive function value y_elite, learn-
ing rate α, learning rate multiplier
c, and parameter ϵ.

In addition to population-based learning, probabilistic prototype trees can also
explore the design space via mutations. The tree is mutated to explore the region

around xbest. Let p be a probability vector in a node that was accessed when
generating xbest. Each component in p is mutated with a probability proportional
to the problem size:

pmutation
#p
√

#xbest
(20.4)

where pmutation is a mutation parameter, #p is the number of components in p,
and #xbest is the number of rules applied in xbest. A component i, selected for
mutation, is adjusted according to:

pi ← pi + β · (1− pi) (20.5)

where β controls the amount of mutation. Small probabilities undergo larger
mutations than do larger probabilities. All mutated probability vectors must be
renormalized. Mutation is implemented in algorithm 20.15 and visualized in
figure 20.10.

function mutate!(ppt, grammar, x_best, p_mutation, β;
sqrtlen = sqrt(length(x_best)),
)
typ = return_type(grammar, x_best)
p = ppt.ps[typ]
prob = p_mutation/(length(p)*sqrtlen)
for i in 1 : length(p)

if rand() < prob
p[i] += β*(1-p[i])

end
end
normalize!(p, 1)
for (pptchild,xchild) in zip(ppt.children, x_best.children)

mutate!(pptchild, grammar, xchild, p_mutation, β,
sqrtlen=sqrtlen)

end
return ppt

end

Algorithm20.15. Amethod formu-
tating a probabilistic prototype tree
with root ppt, grammar grammar,
best expression x_best, mutation
parameter p_mutation, and muta-
tion rate β.

Finally, subtrees in the probabilistic prototype tree are pruned in order to
remove stale portions of the tree. A child node is removed if its parent contains
a probability component above a specified threshold such that, when chosen,
causes the child to be irrelevant. This is always the case for a terminal and may

1

0 probability vector entries

pr
ob

ab
ili
ty

probability vector entries probability vector entries

increase normalize

Figure 20.10. Mutating a prob-
ability vector in a probabilistic
prototype tree with β = 0.5.
The mutated components were
increased according to equa-
tion (20.5) and the resulting
probability vector was renor-
malized. Notice how smaller
probabilities receive greater
increases.

be the case for a nonterminal. Pruning is implemented in algorithm 20.16 and
demonstrated in example 20.7.

20.6 Summary

• Expression optimization allows for optimizing tree structures that, under a
grammar, can express sophisticated programs, structures, and other designs
lacking a fixed size.

• Grammars define the rules used to construct expressions.

• Genetic programming adapts genetic algorithms to perform mutation and
crossover on expression trees.

• Grammatical evolution operates on an integer array that can be decoded into
an expression tree.

• Probabilistic grammars learn which rules are best to generate, and probabilistic
prototype trees learn probabilities for every iteration of the expression rule
generation process.

function prune!(ppt, grammar; p_threshold=0.99)
kmax, pmax = :None, 0.0
for (k, p) in ppt.ps

pmax′ = maximum(p)
if pmax′ > pmax

kmax, pmax = k, pmax′
end

end
if pmax > p_threshold

i = argmax(ppt.ps[kmax])
if isterminal(grammar, i)

empty!(ppt.children[kmax])
else

max_arity_for_rule = maximum(nchildren(grammar, r) for
r in grammar[kmax])

while length(ppt.children) > max_arity_for_rule
pop!(ppt.children)

end
end

end
for child in ppt.children

prune!(child, grammar, p_threshold=threshold)
end
return ppt

end

Algorithm 20.16. A method for
pruning a probabilistic prototype
tree with root ppt, grammar
grammar, and pruning probability
threshold p_treshold.

Consider a node with a probability vector over the rule set:

R 7→ R + R

R 7→ ln(R)

R 7→ 2 | x

R 7→ S

If the probability of selecting 2 or x grows large, then any children in the
probabilistic prototype tree are unlikely to be needed and can be pruned.
Similarly, if the probability of choosing S grows large, any children with
return type R are unneeded and can be pruned.

Example 20.7. An example of
when pruning for probabilistic pro-
totype trees should be applied.

20.7 Exercises

Exercise 20.1. How many expression trees can be generated using the following
grammar and the starting set {R, I, F}?

R 7→ I | F

I 7→ 1 | 2

F 7→ π

Exercise 20.2. The number of expression trees up to height h that can be generated
under a grammar grows super-exponentially. As a reference, calculate the number
of expressions of height h can be generated using the grammar:16 16 Let an empty expression have

height 0, the expression {} have
height 1, and so on.N 7→ {N, N} | {N, } | {, N} | {} (20.6)

Exercise 20.3. Define a grammar which can generate any nonnegative integer.

Exercise 20.4. How do expression optimization methods handle divide-by-zero
values or other exceptions encountered when generating random subtrees?

Exercise 20.5. Consider an arithmetic grammar such as:

R 7→ x | y | z | R + R | R−R | R×R | R/R | ln R | sin R | cos R

Suppose the variables x, y, and z each have units, and the output is expected to
be in particular units. How might such a grammar be modified to respect units?

Exercise 20.6. Consider the grammar

S 7→ NP VP

NP 7→ ADJ NP | ADJ N

VP 7→ V ADV

ADJ 7→ a | the | big | little | blue | red
N 7→ mouse | cat | dog | pony
V 7→ ran | sat | slept | ate

ADV 7→ quietly | quickly | soundly | happily

What is the phenotype corresponding to the genotype [2, 10, 19, 0, 6] and the
starting symbol S?

Exercise 20.7. Use genetic programming to evolve the gear ratios for a clock. As-
sume all gears are restricted to have radii selected fromR = {10, 25, 30, 50, 60, 100}.
Each gear can either be attached to its parent’s axle, thereby sharing the same
rotation period, or be interlocked on its parent’s rim, thereby having a rotation
period depending on the parent’s rotation period and on the gear ratio as shown
in figure 20.11.

The clock can also contain hands, which are mounted on the axle of a parent
gear. Assume the root gear turns with a period of troot = 0.1 s and has a radius of
25. The objective is to produce a clock with a second, minute, and hour hand.

rp

rc

tc = tp
rc
rp

Figure 20.11. The rotation period
tc of a child gear attached to the
rim of a parent gear depends on
the rotation period of the parent
gear, tp and the ratio of the gears’
radii.

Score each individual according to:

(

minimize
hands

(1− thand)
2

)

+

(

minimize
hands

(60− thand)
2

)

+

(

minimize
hands

(3600− thand)
2

)

+ #nodes · 10−3

where thand is the rotation period of a particular hand in seconds and #nodes is
the number of nodes in the expression tree. Ignore rotation direction.

Exercise 20.8. The four 4s puzzle17 is a mathematical challenge in which we use 17 W.W.R. Ball, Mathematical Recre-
ations and Essays. Macmillan, 1892.four 4 digits and mathematical operations to generate expressions for each of the

integers from 0 to 100. For example, the first two integers can be produced by
4 + 4− 4− 4 and 44/44, respectively. Complete the four 4s puzzle.

Exercise 20.9. Consider the probabilistic grammar

R 7→ R + R | R×R | F | I wR = [1, 1, 5, 5]

F 7→ 1.5 | ∞ wF = [4, 3]

I 7→ 1 | 2 | 3 wI = [1, 1, 1]

What is the generation probability of the expression 1.5 + 2?

Exercise 20.10. What is the probabilistic grammar from the previous question
after clearing the counts and applying a learning update on 1.5 + 2?

